*** Do-file to replicate 
*** Manuscript: Time for a Haircut: Political Regimes and Sovereign Debt Restructurings
*** Author: Ignacio Mamone 
*** Journal: International Interactions
*** Date: March 3, 2020
*** Analysis run on StataSE 16


** Load the dataset
clear
use "time_for_a_haircut.dta"

** Prepare event history analysis
stset totdelay, origin(default2==1) failure (res==1)

********************************************
************** TABLE 1: DURATION 
********************************************
* Model 1: Cox. Treatment: polyarchy level
stcox e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew, nohr
est sto a
* Model 2: Lognormal selection-corrected. Treatment: polyarchy level
dursel totdelay e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto b
sum res if e(sample) & res==1
* Model 3: Lognormal selection-corrected Mechanism: Deliberation 
dursel totdelay e_v2x_polyarchy_4C v2xdl_delib cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew, ///
   select(e_v2x_polyarchy_4C v2xdl_delib cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto c
sum res if e(sample) & res==1
* Model 4: Lognormal selection-corrected Mechanism: Transparency   
dursel totdelay e_v2x_polyarchy_4C transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto d
sum res if e(sample) & res==1
* Model 5: Lognormal selection-corrected Mechanism: Checks
dursel totdelay e_v2x_polyarchy_4C checks cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
  select(e_v2x_polyarchy_4C checks cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)  time 
est sto e 
sum res if e(sample) & res==1
* Model 6: Lognormal selection-corrected Mechanism: Election year
dursel totdelay e_v2x_polyarchy_4C electionyear cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
  select(e_v2x_polyarchy_4C electionyear cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)   time
est sto f
sum res if e(sample) & res==1
* Model 7: Lognormal selection-corrected Mechanism Turnover  
dursel totdelay e_v2x_polyarchy_4C turnover cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C turnover cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto g
sum res if e(sample) & res==1

** Obtain results table. Note: when you first open the .rtf table, the stored results for the outcome "restructuring duration" for Models 2 through 7 are fliped in relation to what is shown in Table 1 in the manuscript, i.e. the results for M2-M7 in the upper panel of the rtf table correspond to the outcome "selection into restructuring" but to facilitate interpretation for readers you should manually move those results to the bottom panel so that the final table (as printed in the manuscript) shows in its upper panel the results for the outcome "restructuring duration".
esttab a b c d e f g using "Table1.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01) 
   
   
*********************************************
******** TABLE 2: REESTRUCTURINGS' HAIRCUTS
*********************************************
*** Create constrained variables for hurdle double tobit estimation
gen haircutbounded = haircut
replace haircutbounded=0.1 if haircut<=0
gen haircutboundednew = haircutbounded/100
replace haircutboundednew=0 if haircutbounded==.
label var haircutboundednew "Haircut (% creditor loses)"
gen bondnew = 0
replace bondnew= 1 if bond_restructuring==1 
gen HIPCnew = 0
replace HIPCnew = 1 if HIPC_IDA==1 

* Model 1: Treatment: polyarchy level
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew ) vce(cluster cow)
est sto a
sum rescase if e(sample) & rescase==1
* Model 2: Mechanism: Deliberation
craggit rescase e_v2x_polyarchy_4C v2xdl_delib cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C v2xdl_delib cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew ) vce(cluster cow)
est sto b
sum rescase if e(sample) & rescase==1
* Model 3: Mechanism: Transparency
craggit rescase e_v2x_polyarchy_4C transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew ) vce(cluster cow)
est sto c 
sum rescase if e(sample) & rescase==1
* Model 4: Mechanism: Checks
craggit rescase e_v2x_polyarchy_4C checks cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C checks cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew ) vce(cluster cow)
est sto d
sum rescase if e(sample) & rescase==1
* Model 5: Mechanism: Election year
craggit rescase e_v2x_polyarchy_4C electionyear cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C electionyear cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew ) vce(cluster cow)
est sto e
sum rescase if e(sample) & rescase==1
* Model 6: Mechanism: Turnover
craggit rescase e_v2x_polyarchy_4C turnover cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C turnover cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew ) vce(cluster cow)
est sto f
sum rescase if e(sample) & rescase==1

esttab a b c d e f using "Table2.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01) 


** Create Figure#2 of Predicted Haircuts
gen polyarchy = .
replace polyarchy = 1 if v2x_polyarchy>.05 | v2x_polyarchy<.35 & v2x_polyarchy!=.
replace polyarchy = 2 if v2x_polyarchy>=.35 | v2x_polyarchy<.75 & v2x_polyarchy!=.
replace polyarchy = 3 if v2x_polyarchy>=.75 & v2x_polyarchy!=.
regress haircut polyarchy cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew, vce(cluster cow)
margins, at(polyarchy = (1 3)) atmeans 
marginsplot


************************************************
*********** APPENDICES AND ROBUSTNESS  *********

** Figure S.1 Histogram of duration
histogram totdelay if res==1, bin(26) percent fcolor(gs13) lcolor(gs10) lwidth(vthin) kdensity ///
 kdenopts(lcolor(black)) ylabel(0(5)30, labsize(small) nogrid) xlabel(0(25)275, labsize(small)) ///
 graphregion(fcolor(white) lcolor(white) ifcolor(white) ilcolor(white))

** Figure S.2 Histogram of haircuts
histogram haircut if haircut!=., bin(20) percent fcolor(gs13) lcolor(gs10) lwidth(vthin) kdensity ///
 kdenopts(lcolor(black)) ylabel(0(5)12, labsize(small) nogrid) xlabel(-10(10)100, labsize(small)) ///
 graphregion(fcolor(white) lcolor(white) ifcolor(white) ilcolor(white))

** Table S.2.a Descriptive Statistics (Reestructurings)
sum totdelay  e_v2x_polyarchy_4C  checks turnover electionyear v2xdl_delib transparencyindex v2x_polyarchy polity dpi17parliam v2lgtreaty dpi17left  gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew us10yrrate   currencycrises  HIPC_IDA preemptivenew yrslast3 if res!=. & totdelay!=.

** Table S.2.b Descriptive Statistics (Haircuts)
sum haircut e_v2x_polyarchy_4C  checks turnover electionyear v2xdl_delib transparencyindex v2x_polyarchy polity dpi17parliam v2lgtreaty dpi17left  gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew us10yrrate   currencycrises  HIPC_IDA preemptivenew yrslast3 if rescase==1 & haircut!=.

** Table S.2.c Descriptive Statistics (Full sample)
sum default2 e_v2x_polyarchy_4C  checks turnover electionyear v2xdl_delib transparencyindex v2x_polyarchy polity dpi17parliam v2lgtreaty dpi17left  gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents stratanew us10yrrate   currencycrises  HIPC_IDA preemptivenew yrslast3 if default2!=.


****** TABLE S.3: Event History Analysis Robustness I
* Reduced covariates
dursel totdelay e_v2x_polyarchy_4C debt_gdp gdppclog , select(e_v2x_polyarchy_4C debt_gdp gdppclog) dist(lognormal) cluster(cow)
est sto a
* POLYARCHY CONTINUOUS
dursel totdelay v2x_polyarchy cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(v2x_polyarchy cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto b
* POLYARCHY HIGH CODE
dursel totdelay v2x_polyarchy_codehigh cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(v2x_polyarchy_codehigh cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto c
* POLYARCHY LOW CODE
dursel totdelay v2x_polyarchy_codelow cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(v2x_polyarchy_codelow cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto d
* POLITY
dursel totdelay polity cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(polity cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto e
* All mechanisms together
dursel totdelay e_v2x_polyarchy_4C checks electionyear turnover v2xdl_delib transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C checks electionyear turnover v2xdl_delib transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time  
est sto f
* All autocracies together   
dursel totdelay gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew, ///
select(gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) ///
dist(lognormal) cluster(cow) time
est sto g
** Obtain results table. Note: when you first open the .rtf table, the stored results for the outcome "restructuring duration" are fliped in relation to what is shown in Table S.3, i.e. the results in the upper panel of the rtf table correspond to the outcome "selection into restructuring" but to facilitate interpretation for readers you should manually move those results to the bottom panel so that the final table (as printed in the supplementary materials) shows in its upper panel the results for the outcome "restructuring duration".
esttab a b c d e f g using "TableS3.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01)


****** TABLE S.4: Event History Analysis Robustness II 
* PREEMPTIVE RESTRUCTURING
dursel totdelay e_v2x_polyarchy_4C cbi preemptivenew debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C cbi preemptivenew debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto a
* CURRENCY CRISES
dursel totdelay e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports currencycrises resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports currencycrises resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto b
* Regional Debt/GDP
dursel totdelay e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents regionavg_debtGDP) dist(lognormal) cluster(cow)
est sto c
* HIPC
dursel totdelay e_v2x_polyarchy_4C cbi HIPC_IDA debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto d
* Years since last res.^3
dursel totdelay e_v2x_polyarchy_4C cbi yrslast3 debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring, ///
   select(e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow)
est sto e
** Obtain results table. Note: when you first open the .rtf table, the stored results for the outcome "restructuring duration" are fliped in relation to what is shown in Table S.4, i.e. the results in the upper panel of the rtf table correspond to the outcome "selection into restructuring" but to facilitate interpretation for readers you should manually move those results to the bottom panel so that the final table (as printed in the supplementary materials) shows in its upper panel the results for the outcome "restructuring duration".
esttab a b c d e  using "TableS4.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01)   


****** TABLE S.5: Double Hurdle Models of Haircuts: Robustness I
* Reduced covariates
craggit rescase e_v2x_polyarchy_4C debt_gdp gdppclog us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C debt_gdp gdppclog) 
est sto a
* Polyarchy cont.
craggit rescase v2x_polyarchy cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew v2x_polyarchy cbi debt_gdp gdppclog exports fixedrate resourcerents) vce(cluster cow)
est sto b
* Polyarchy codehigh
craggit rescase v2x_polyarchy_codehigh cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew v2x_polyarchy_codehigh cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto c
* Polyarchy codelow
craggit rescase v2x_polyarchy_codelow cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew v2x_polyarchy_codelow cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto d
* Polity
craggit rescase polity cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew polity cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto e
* All mechanisms together
craggit rescase e_v2x_polyarchy_4C checks electionyear turnover v2xdl_delib transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C checks electionyear turnover v2xdl_delib transparencyindex cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto f
* All autocracies together
craggit rescase gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew gwf_party gwf_military gwf_monarchy gwf_personal cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto g
esttab a b c d e f g using "TableS5.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01)  


****** TABLE S.6: Double Hurdle Models of Haircuts: Robustness II
* Excludes bonds
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate if bondnew==0, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents) vce(cluster cow)
est sto a
* Excludes preemptive
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate if preemptivenew==0, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents) vce(cluster cow)
est sto b
* Excludes HIPC
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate if HIPCnew==0, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto c
* Excludes HIPC (and CBI)
craggit rescase e_v2x_polyarchy_4C debt_gdp gdppclog exports fixedrate resourcerents us10yrrate if HIPCnew==0, ///
 second(haircutboundednew e_v2x_polyarchy_4C debt_gdp gdppclog exports fixedrate resourcerents )  vce(cluster cow)
est sto d
* Currency crises
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports currencycrises resourcerents  us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports currencycrises resourcerents ) vce(cluster cow)
est sto e
* Regional Debt/GDP
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents  regionavg_debtGDP, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto f
* Years since last^3
craggit rescase e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents yrslast3 us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C cbi debt_gdp gdppclog exports fixedrate resourcerents yrslast3 ) vce(cluster cow)
est sto g
esttab a b c d e f g using "TableS6.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01)  
 
 
****** TABLE S.7 VARIETIES OF DEMOCRACY      
*********** Panel A: Duration 
* Polyarchy X Parliamentary
dursel totdelay e_v2x_polyarchy_4C dpi17parliam intpolyparliam cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C dpi17parliam intpolyparliam cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto a
sum res if e(sample) & res==1  
* Polyarchy X Ratification
dursel totdelay e_v2x_polyarchy_4C v2lgtreaty intpolytreaty cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew ,  ///
   select(e_v2x_polyarchy_4C v2lgtreaty intpolytreaty cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto b
sum res if e(sample) & res==1  
* Polyarchy X Partisanship
dursel totdelay e_v2x_polyarchy_4C dpi17left intpolyleft cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew , ///
   select(e_v2x_polyarchy_4C dpi17left intpolyleft cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto c
sum res if e(sample) & res==1  

*********** Panel B: Haircuts 
* Polyarchy X Parliamentary
craggit rescase e_v2x_polyarchy_4C dpi17parliam intpolyparliam cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C dpi17parliam intpolyparliam cbi debt_gdp gdppclog exports fixedrate resourcerents) vce(cluster cow)
est sto d
* Polyarchy X Ratification
craggit rescase e_v2x_polyarchy_4C v2lgtreaty intpolytreaty cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C v2lgtreaty intpolytreaty cbi debt_gdp gdppclog exports fixedrate resourcerents)  vce(cluster cow)
est sto e
* Polyarchy X Partisanship
craggit rescase e_v2x_polyarchy_4C dpi17left intpolyleft cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew e_v2x_polyarchy_4C dpi17left intpolyleft cbi debt_gdp gdppclog exports fixedrate resourcerents )  vce(cluster cow)
est sto f
esttab a b c d e f using "TableS7.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01) 


****** TABLE S.8 VARIETIES OF AUTOCRACY 
***** Panel A: Duration
dursel totdelay gwf_party cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew, ///
select(gwf_party cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time
est sto a   
dursel totdelay gwf_military cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew, ///
select(gwf_military cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time   
est sto b
dursel totdelay gwf_monarchy  cbi debt_gdp gdppclog exports fixedrate resourcerents bond_restructuring stratanew, ///
select(gwf_monarchy cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate) dist(lognormal) cluster(cow) time  
est sto c

***** Panel B: Haircuts
craggit rescase gwf_party cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew gwf_party cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto d
craggit rescase gwf_military cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew gwf_military cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto e
craggit rescase gwf_monarchy cbi debt_gdp gdppclog exports fixedrate resourcerents us10yrrate, ///
 second(haircutboundednew gwf_monarchy cbi debt_gdp gdppclog exports fixedrate resourcerents ) vce(cluster cow)
est sto f
esttab a b c d e f using "TableS8.rtf", label replace nodepvars se(3) b(3) compress nogaps star(* 0.10 ** 0.05 *** 0.01)
